

Fitting Compartment Models Using PROC NL MIXED

Raghavendra Rao Kurada, SAS Institute Inc.
Fang Chen, SAS Institute Inc.

ABSTRACT

The CMPTMODEL statement is a new enhancement to the NL MIXED procedure in SAS/STAT[®] 14.3. This statement enables you to fit a large class of pharmacokinetics (PK) models, including one-, two-, and three-compartment models, with intravenous (bolus and infusion) and extravascular (oral) types of drug administration. The CMPTMODEL statement also supports multiple dosages and PK models that have various parameterizations. This paper introduces the new statement and illustrates its usage through examples. Related concepts are also discussed, such as the %PKCONVRT autocall macro (which converts PK data sets that are stored according to industry standard to data sets that can be directly used by PROC NL MIXED), extension to E_{max} models, prediction, visualization, and fitting Bayesian PK models (by using the MCMC procedure).

Introduction

Pharmacokinetics (PK) is a branch of medicine that studies the movement of drugs in the body. In clinical trials, a dosage of a drug is given to a patient, either intravenously or extravascularly, and drug concentration levels are measured in the follow-up time period. PK models use drug dosage, lapsed time, and additional covariates to model the concentrations. In a structural PK model, the drug concentrations are characterized by a set of ordinary differential equations, and the solutions of these equations are further used to model an often nonlinear relationship between the drug concentrations and time.

A large class of PK models are derived on the basis of the *compartment* notion, where in each compartment, the behavior of a drug is kinetically homogeneous. The most common and basic compartment models are the one-, two-, and three-compartment models. Solutions to the differential equations that correspond to these compartment models (for intravenous and extravascular types of drug administration) are known; see Abuhelwa, Foster, and Upton (2015) and Fisher and Shafer (2007). The new CMPTMODEL statement in SAS/STAT 14.3 simplifies the specification of these compartment models and provides analytical solutions to predicted concentrations in order to speed up the estimation process.

This paper is divided into five main sections. The first section provides background on the compartment models that are handled by the new CMPTMODEL statement. The second section introduces the CMPTMODEL statement. The third and fourth sections explain how to fit PK models to multiple-dosage data. The fifth section shows how to fit Bayesian PK models by using the CMPTMODEL statement in the MCMC procedure.

Compartment Models

Pharmacokinetics studies the movement of drugs in the body. Detailed mathematical modeling of several pharmacokinetics studies are given by Gabrielsson and Weiner (2006), and a few basic concepts from this reference are discussed in this section. The movement of a drug within the body is described by the following process, abbreviated as ADME:

absorption: describes how the drug enters the body (intravenously or extravascularly)

distribution: describes how the drug is distributed through the various components of the body

metabolism: describes how the drug compound breaks down into other compounds (known as metabolites)

excretion: describes how the drug and its metabolites are removed from the body

During the ADME process, the drug concentration at any given time is of interest. Models that describe the behavior of drug concentrations are known as pharmacokinetic (PK) models. Broadly, PK models are divided into two types: structural PK models and covariate PK models. Structural PK models describe the relationship between the concentrations and time and are commonly derived by using a set of differential equations. Covariate PK models describe the relationship between the concentrations and the other covariates. In practice, you create an appropriate structural PK model by fitting several structural PK models for a given set of data, and then you combine this model with a covariate PK model to come up with a final PK model. For more information about structural and covariate PK models, see Mould and Upton (2012) and Bonate (2011).

Compartment models are the basic building blocks of these PK models. A *compartment* is a group of organs or tissues in a body that are kinetically homogeneous. The body is divided into a system of (hypothetical) compartments. From the time a drug enters the body, it is distributed through several compartments before it is eliminated from the body (ADME process). Compartment models describe the relationship between the drug concentration and time and other covariates in a compartment. The drug concentration in a compartment at any given time is directly related to the amount of that drug present in the compartment. You characterize the amount of drug in a compartment by using a set of differential equations (Gabrielsson and Weiner 2006). In general, these differential equations are not analytically solvable for all multicompartment models. However, solutions of the one-, two- and three-compartment models are available in Abuhelwa, Foster, and Upton (2015) and Fisher and Shafer (2007), and a new feature in the NL MIXED procedure uses these solutions to fit one-, two-, and three-compartment models. This section explains the one-, two-, and three-compartment models in detail; this paper restricts its scope to these models.

Assumptions

In compartment models, the assumption is that there is always a *central* compartment, and there might be one or more *peripheral* compartments. Also, it is assumed that the drug is administered at a specific site (such as the arm) and is distributed only through the central compartment to other, peripheral compartments (see Figure 1 and Figure 2). Sometimes the central compartment is the administration site itself. If not, the administration site is referred to as a *depot* compartment. In this case, the drug first moves from the depot compartment to the central compartment and is then distributed to the peripheral compartments. The rate at which the drug moves from the depot compartment to the central compartment is known as the absorption rate; it is characterized by the absorption rate constant. Similarly, the rate at which the drug moves from the central compartment to and from the peripheral compartments is known as the transfer rate; it is characterized by the transfer rate constant. The rate at which the drug is eliminated from the central or peripheral compartments and the rate at which the drug is eliminated from a given compartment are known as the elimination rate and characterized by the elimination rate constant. On the basis of these absorption, transfer, and elimination rate constants and the initial dosage of the drug, you can model the amount of drug present in a compartment at any given time.

Routes of Drug Administration


Another major concept in compartment modeling is the method by which a drug is administered to a patient. A drug can be administered either intravenously or extravascularly. The most common drug administration types that are considered in PK models are intravenous bolus, intravenous infusion, and extravascular oral dose. A bolus medication typically has no time lag or a very short time lag for the drug to enter the central compartment. Intravenous infusions are administered over a period of time, and the drug enters the body at a constant rate. In infusion, two quantities—the rate and duration of the infusion—are used to calculate the amount of the drug present in the body. The rate and duration of the infusion are related, so knowing one enables you to know the other. For compartment models with an infusion type of administration, it is sufficient to provide either the rate or the duration. Extravascular administrations, such as oral administration, differ from bolus and infusion in the sense that there is an absorption phase before the drug enters the central compartment.

One-, Two-, and Three-Compartment Models of Intravenous Drug Administration

Figure 1 shows schematic representations of the one-, two-, and three-compartment models of intravenous drug administration. In each diagram, *Dose* is the initial dosage of the drug that is given intravenously, and the drug is eliminated from compartment 1 at a rate of k_{10} . For two- and three-compartment models, k_{12} is the transfer rate constant from compartment 1 to compartment 2, and k_{21} is the transfer rate constant from compartment 2 to compartment 1. Similarly, for the three-compartment model, k_{13} is the transfer rate constant from compartment 1 to compartment 3, and k_{31} is the transfer rate constant from compartment 3 to compartment 1. Compartment 1 in all

the compartment models is considered to be the central compartment, and compartment 2 and compartment 3 are considered to be peripheral compartments.


Figure 1 Schematic Diagrams of Compartment Models of Intravenous Drug Administration



One-, Two-, and Three-Compartment Models of Extravascular Drug Administration

Unlike intravenous models, extravascular models have an absorption phase. The most common type of extravascular drug administration is oral administration. Figure 2 shows schematic representations of the one-, two-, and three-compartment models of extravascular drug administration. In these models, the drug is administered in a depot compartment, which is represented as compartment 0 in each diagram. Also, k_a represents the absorption rate constant, which characterizes the absorption rate at which the drug moves from the depot compartment to the central compartment.

Figure 2 Schematic Diagrams of Compartment Models of Extravascular Drug Administration



Compartment Models and Differential Equations

On the basis of the setup described in the preceding section, you can calculate the amount of drug in each compartment by using a set of differential equations for each compartment model. For example, for the one-compartment model with an intravenous bolus dose administration, the rate at which the drug moves in the central compartment can be calculated by the following differential equation,

$$\frac{dA_1(t)}{dt} = -k_{10} \times A_1$$

where $A_1(t)$ is the amount of the drug in compartment 1 at time t . Similarly, for a one-compartment model with an oral drug administration, the set of differential equations can be written as

$$\begin{aligned} \frac{dA_0(t)}{dt} &= -k_a \times A_0 \\ \frac{dA_1(t)}{dt} &= (k_a \times A_0) - (k_{10} \times A_1) \end{aligned}$$

where $A_0(t)$ is the amount of the drug in compartment 0 at time t . Solutions to these differential equations give you the relationship between the amount of drug in a compartment and time. For example, the solution to the first equation in this section when the initial condition $A_1(0) = D$ is given as

$$A_1(t) = D \times \exp(-k_{10} \times t)$$

Solutions to the two- and three-compartment models are more complex and are presented in the section “[Solutions to the One-, Two-, and Three-Compartment Model Differential Equations](#)” on page 21 in the Appendix for both the intravenous and extravascular drug administration types. These results are used in a new feature in PROC NL MIXED to fit one-, two-, and three-compartment models by using a simple syntax.

CMPTMODEL Statement in PROC NL MIXED

The CMPTMODEL statement in the NL MIXED procedure consists of options and combinations of options that enable you to fit one-, two-, and three-compartment models. Although the total number of options is large, they can be divided into the following three categories to help you understand how they work:

- Required options: PCONC=, TIME=,¹ NCOMPS=, ADMTYPE=, and PARMTYPE=
 - Specify the outcome (drug concentration) variable, the time variable, and the type of compartment models that you want to fit
- Conditionally required options: CL1=, CL2=, CL3=, VOL1=, VOL2=, VOL3=, K12=, K21=, K13=, K31=, KA=, K10=, K20=, K30=, RATE=, and DURN=
 - Specify the particular requirements of the compartment model that you request
- Non-required options: DOSE0=, DOSE1=, DOSE2=, DOSE3=, SCALE0=, SCALE1=, SCALE2=, SCALE3=, PCONC0=, PCONC2=, and PCONC3=

The CMPTMODEL statement supports nine types of compartment models, which you must specify:

- one-compartment model (NCOMPS=1) with bolus dose administration (ADMTYPE=BOLUS)
- one-compartment model (NCOMPS=1) with infusion administration (ADMTYPE=INFUSION)
- one-compartment model (NCOMPS=1) with oral administration (ADMTYPE=ORAL)
- two-compartment model (NCOMPS=2) with bolus dose administration (ADMTYPE=BOLUS)
- two-compartment model (NCOMPS=2) with infusion administration (ADMTYPE=INFUSION)
- two-compartment model (NCOMPS=2) with oral administration (ADMTYPE=ORAL)
- three-compartment model (NCOMPS=3) with bolus dose administration (ADMTYPE=BOLUS)
- three-compartment model (NCOMPS=3) with infusion administration (ADMTYPE=INFUSION)
- three-compartment model (NCOMPS=3) with oral administration (ADMTYPE=ORAL)

Each model can be parameterized in terms of either elimination and transfer rate constants (PARMTYPE=1) or clearance and volume parameters (PARMTYPE=2). You must specify the parameterization type in the statement. In total, you can use the CMPTMODEL statement to fit 18 types of compartment models.

The PARMTYPE=1 models require additional specification of the K_{ij} = (rate) options. The PARMTYPE=2 models require additional specification of the CL_n = (clearance) and VOL_n = (volume) options. The relationship between the rate constants and the clearance and volumes of each compartment is as follows:

$$K_{10} = CL1/VOL1$$

$$K_{12} = CL2/VOL1$$

$$K_{13} = CL3/VOL1$$

$$K_{21} = CL2/VOL2$$

$$K_{31} = CL3/VOL3$$

Here CL_n and VOL_n are the clearance and volume, respectively, of the n th numbered compartment.

The list of all conditionally required and non-required options is given in the section “[CMPTMODEL Statement and Its Options](#)” on page 25 of the Appendix.

¹The TIME= option is not required if you use a converted data set by running the %PKCONVRT autocall macro on an industry-standard data set. See the Theophylline example in the section “[%PKCONVRT Autocall Macro and the CMPTMODEL Statement](#)” on page 10.

Scaling

The amount of drug and the drug concentration in a compartment are directly related: the drug concentration is the drug amount divided by the apparent volume of the drug in a compartment. However, often the concentration and the amount of drug administered are measured in different units. As a result, the apparent volume of the compartment units needs to be scaled appropriately. The $SCALE_n$ option (for the n th compartment) specifies the scaling factor that is used to divide the predicted concentrations. By default, $SCALE_n=1$.

Theophylline Example

This section illustrates how to fit a simple one-compartment model to the well-known theophylline drug data set (Pinheiro and Bates 1995). In this experiment, 12 patients are given a dosage of the anti-asthmatic drug theophylline orally, and the serum concentrations in their blood are measured for a 25-hour time period. A partial data set follows:

```
data Theoph;
  input patient time conc dose;
  datalines;
1 0.00 0.74 4.02
1 0.25 2.84 4.02
1 0.57 6.57 4.02
1 1.12 10.50 4.02
1 2.02 9.66 4.02
1 3.82 8.58 4.02
1 5.10 8.36 4.02
1 7.03 7.47 4.02
... more lines ...
12 24.15 1.17 5.30
;
```

The Theoph data set includes the following variables:

- patient: patient identifier
- time: time at which the serum concentration was measured
- conc: serum concentration
- dose: initial dose given to the patient

The kinetics of theophylline is analyzed by Pinheiro and Bates (1995), Vonesh (2012), and many others. In order to analyze the relationship between the serum concentrations and time, Pinheiro and Bates (1995) suggested a one-compartment model with an oral dose administration. You can fit this model by using the following program:

```
proc nlmixed data=Theoph;
  parms beta1=-3.22 beta2=0.47 beta3=-2.45
        s2b1 =0.03 cb12 =0 s2b2 =0.4 s2=0.5;
  c1 = exp(beta1 + b1);
  ka = exp(beta2 + b2);
  ke = exp(beta3);
  v1 = c1/ke;
  CMPTMODEL pconc=predConc time=time ncomps=1 admtype=oral parmtime=1
            k10=ke ka=ka
            dose0=dose scale1=v1;
  /* analytic solution */
  /* predConc = dose*ke*ka*(exp(-ke*time)-exp(-ka*time))/c1/(ka-ke); */
  model conc ~ normal(predConc, s2);
  random b1 b2 ~ normal([0,0],[s2b1,cb12,s2b2]) subject=patient;
run;
```

In this CMPTMODEL statement, the variable that is specified in the PCONC= option (predConc) is the predicted concentration at each time point (specified in the TIME= option). The predConc variable is used in the likelihood function as the mean of the normal distribution. The NCOMPS=1 and ADMTYPE=ORAL options request a one-compartment model with an oral administration type. The PARMTYPE=1 option parameterizes the model in terms of the elimination rate constant, which requires the specification of the rate K10= option. In addition, the oral administration type requires the specification of the absorption rate (KA= option). Finally, the initial dosage in the depot compartment is specified by the DOSE0= option, and the SCALE1=vl option scales the value of predConc by vl.

The PROC NL MIXED program produces the parameter estimates displayed in Figure 3.

Figure 3 Parameter Estimates for One-Compartment Model

Parameter Estimates								
Parameter	Estimate	Standard Error	DF	t Value	Pr > t	95% Confidence Limits		Gradient
beta1	-3.2268	0.05950	10	-54.23	<.0001	-3.3594	-3.0942	-0.00009
beta2	0.4806	0.1989	10	2.42	0.0363	0.03745	0.9238	3.645E-7
beta3	-2.4592	0.05126	10	-47.97	<.0001	-2.5734	-2.3449	0.000039
s2b1	0.02803	0.01221	10	2.30	0.0446	0.000828	0.05524	-0.00014
cb12	-0.00127	0.03404	10	-0.04	0.9710	-0.07712	0.07458	-0.00007
s2b2	0.4331	0.2005	10	2.16	0.0561	-0.01354	0.8798	-6.98E-6
s2	0.5016	0.06837	10	7.34	<.0001	0.3493	0.6540	6.133E-6

The analytic solution of the one-compartment model for an oral dose administration type is known:

$$C(t) = \frac{D \times k_a}{V \times (k_a - k_{10})} \times \{\exp(-k_{10}t) - \exp(k_a t)\}$$

The formula is coded in the program and commented out. If you execute the same program without the CMPTMODEL statement but use the analytic solution, you get identical estimates.

Pharmacokinetics Data Structure and the %PKCONVRT Autocall Macro

In the Theophylline data set, the variable TIME records the elapsed times since the theophylline was administered. This is somewhat atypical. Often, it is the times at measurement that are recorded, not the elapsed times. The CMPTMODEL statement assumes that the variable specified in the TIME= option records the elapsed times, which might be difficult and complex to compute, especially in multiple-dosage scenarios. SAS/STAT 14.3 supports a %PKCONVRT autocall macro that you can use to convert a data set to the format that PROC NL MIXED can use.

The %PKCONVRT autocall macro converts pharmacokinetics data that are stored according to industry convention (Owen and Fiedler-Kelly (2014); used, for example, by NONMEM software) to a data set that you can analyze by using PROC NL MIXED. The autocall macro accounts for scenarios that include single dosing, multiple dosing, and different types of drug administration. To simplify the discussion, let's denote the industry-standard input data set to the %PKCONVRT macro as PK-DataSet. The following sections provide some details about this type of data set.

PK-DataSet: Observations and Variables

Every observation in a PK-DataSet must be an event—dosing event, measurement event, or other type of event—and must be coded accordingly. Observations that record dosage amount and dose administration time are *dosing event* observations; observations that record concentration value and the corresponding time are *measurement event* observations. The other types of event record a change in a covariate, an unconventional dosage of a drug given to a patient, and other rare scenarios. For descriptions of these events, see Owen and Fiedler-Kelly (2014). In general, a large portion of a PK-DataSet should contain dosing event and measurement event observations, along with some observations of other event types.

The following variable names are reserved for a PK-DataSet:

- TIME: the time values when a drug is administered or when a concentration is measured. The TIME variable can take either numeric values or time-formatted values. The numeric values must be coded in hours of the day. For example, 8:00 a.m. is coded as 8.0 and 10:30 p.m. is coded as 22.5.
- EVID: the event indicator. Its value can be 0, 1, 2, 3, or 4. The value 0 indicates a measurement event; the value 1 indicates a dosing event; and the values 2, 3, and 4 represent events of other types. For more information about these values, see Owen and Fiedler-Kelly (2014).
- DV: the concentration measurements
- AMT: the dosages. The values of this variable must be strictly positive.
- ID: the subject or patient indicator
- DATE: the date values at either dosing or measurement time. The DATE variable must follow the mmddyy. format.
- DAT1: similar to DATE; must follow the ddmmyy. format
- DAT2: similar to DATE; must follow the yymmdd. format
- CMT: the number of the compartment in which a dosing event or a measurement event takes place. This variable can take any integer value. For one-, two-, and three-compartment models, the values 1 and 2 have a special meaning that is discussed later in this section.
- RATE: the rate of dosing for the infusion. This variable can take any nonnegative value, -1, or -2. The meanings of these values are discussed later in this section.
- ADDL: the number of additional dosages that are administered at a constant time interval
- SS: steady state dosing and measurement events
- II: the fixed time interval between additional doses. This variable is required if either the ADDL or SS variable is present.

A PK-DataSet must contain the TIME, EVID, DV, and AMT variables and can contain the ID, DATE, DAT1, DAT2, CMT, RATE, ADDL, SS, and II variables along with other covariates or independent variables.

For a dosing event observation (EVID=1), the AMT and TIME values should be the amount of drug and the time when the drug is administered, respectively, and the DV value should be either missing or zero. For a measurement event observation (EVID=0), the DV and TIME values should be the drug concentration and the time when the drug is measured, respectively, and the AMT value should be either missing or zero.

You cannot combine the dosing event and the measurement event into a single observation, even if the drug is administered and the concentration is measured at the same time. You must separate them into two observations.

A simple PK-DataSet example is given in the following table:

ID	AMT	TIME	CONC	EVID	WEIGHT
1	4.02	8:30	.	1	64.8
1	.	9:00	2.84	0	64.8
1	.	10:00	6.57	0	64.8
1	.	11:30	10.5	0	64.8
1	.	12:10	9.66	0	64.8
1	.	13:40	8.58	0	64.8

In this data set, a dose in the amount of 4.02 is given to patient 1 at 8.30 a.m., and a series of concentration measurements are taken at the later times. Weight is a covariate.

Intravenous and Extravascular Dosages

You use the EVID, CMT, and RATE variables to indicate whether a drug is administered intravenously or extravascularly.

- bolus dose: the central compartment is used for both drug administration and concentration measure (Figure 1). You specify CMT=1 for both EVID=0 and EVID=1.
- oral dose: the drug is administered in a depot compartment, and the concentrations are measured from the central compartment (Figure 2). You specify CMT=1 for EVID=1 and CMT=2 for EVID=0.

An example PK-DataSet of a bolus dose and an oral dose follows:

ID	AMT	TIME	CONC	EVID	CMT
1	4.02	8:30	.	1	1
1	.	9:00	2.84	0	1
1	.	10:00	6.57	0	1
1	.	11:30	10.5	0	1
1	.	12:10	9.66	0	1
1	.	13:40	8.58	0	1
2	3.95	8:30	.	1	1
2	.	9:00	2.84	0	2
2	.	10:00	6.57	0	2
2	.	11:30	10.5	0	2
2	.	12:10	9.66	0	2
2	.	13:40	8.58	0	2

A bolus dose is given to patient 1, and an oral dose is given to patient 2. Measurements are taken later.

The RATE variable is used to record the intravenous infusion type of dosing. For dosing event observations, the RATE variable can take the following values:

- RATE > 0: indicates an infusion dose, where the positive value is the rate of infusion
- RATE = 0: indicates a noninfusion (bolus or oral) dose
- RATE = -1: indicates an infusion dose, where the rate of infusion is a parameter in the model
- RATE = -2: indicates an infusion dose, where the infusion process is expressed in terms of duration

For measurement event observations, the RATE variable must be either a missing value or zero.

In the following PK-DataSet, an infusion of a drug (in the amount of 4.02) is given to a patient at a RATE of 2.01:

ID	AMT	TIME	RATE	EVID	CONC
1	4.02	8:30	2.01	1	.
1	.	9:00	.	0	2.84
1	.	10:00	.	0	6.57
1	.	11:30	.	0	10.5
1	.	12:10	.	0	9.66
1	.	13:40	.	0	8.58

Multiple Dosages

Drugs are often given in multiple doses during a clinical trial. Sometimes, a drug can be given multiple times consecutively with no concentration measurements taken between doses, or a drug can be administered at regular (or irregular) intervals with its concentration measured between doses. You can account for these scenarios according to the PK-DataSet convention. An example of multiple dosages follows:

ID	AMT	TIME	CONC	EVID	CMT
1	4.02	7:30	.	1	1
1	3.01	8:00	.	1	1
1	2.00	8:30	.	1	1
1	.	9:00	2.84	0	1
1	.	10:00	6.57	0	1
1	.	11:30	10.5	0	1
1	.	12:10	9.66	0	1
1	.	13:40	8.58	0	1
1	2.00	14:30	.	1	1
1	.	15:00	3.84	0	1
1	.	16:00	6.87	0	1
1	.	17:30	4.50	0	1

Patient 1 is given three continuous doses, at 7:30, 8:00, and 8:30 a.m., with no concentration measured between doses. Starting at 9:00 a.m., the concentration is measured five times before an additional dose is administered at 2:30 p.m.

In multiple-dosage analysis, it is commonly assumed that only consecutive doses have cumulative effects: after a concentration is measured, the effect of any additional dose on the predicted concentration is independent of previous doses. For example, in the previous multiple-dosage example data set, predicted concentrations between 9:00 a.m. and 2:40 p.m. depend on the three doses from 7:00 to 8:30 a.m. But the predicted concentrations from 3:00 to 5:30 p.m. depend on only the single dose administered at 2:30 p.m. However, you can fit a model without this assumption. (See the example in the section “Phenobarbital Experiment” on page 11.)

You can also use the variable ADDL to indicate multiple-continuous dosing at a constant time interval. ADDL specifies the number of additional doses to be administered, and the variable II (which is required if ADDL appears in the data set) specifies the time interval at which these additional doses are given.

Consider the following example data set:

ID	AMT	TIME	CONC	EVID	CMT	ADDL	II
1	4.02	7:30	.	1	1	2	0:30
1	.	9:00	2.84	0	1	.	.
1	.	10:00	6.57	0	1	.	.
1	.	11:30	10.5	0	1	.	.
1	.	12:10	9.66	0	1	.	.
1	.	13:40	8.58	0	1	.	.
1	2.00	14:30	.	1	1	.	.
1	.	15:00	3.84	0	1	.	.
1	.	16:00	6.87	0	1	.	.
1	.	17:30	4.50	0	1	.	.

A bolus dose of 4.02 is given at 7:30 a.m., and two additional doses (ADDL=2) in the amount of 4.02 are given at intervals of 30 minutes (II=0:30) after the initial dose.

For additional scenarios that represent different experiment settings in multiple-dosage regimens that use the PK-DataSet convention, see Owen and Fiedler-Kelly (2014).

%PKCONVRT Autocall Macro

The CMPTMODEL statement cannot be directly used to fit a PK model to a PK-DataSet. You must first use the %PKCONVRT macro to convert the data set to another data set that is suitable to use in PROC NL MIXED.

The %PKCONVRT autocall macro takes an input data set (a PK-DataSet), modifies it, and creates an output data set. The output data set can be used by PROC NL MIXED to fit a single-dose or multiple-dose model. In the following DATA step, Theoph_pk is the same theophylline data set except that it is stored according to the PK-DataSet convention. Each subject has a dosing event observation that is followed by measurement event observations. The EVID variable indicates event types.

```
data Theoph_pk;
  input ID TIME DV AMT EVID;
datalines;
  1  0.00  .  4.02  1
  1  0.00  0.74  .  0
  1  0.25  2.84  .  0
  1  0.57  6.57  .  0
  1  1.12 10.50  .  0
  1  2.02  9.66  .  0
  1  3.82  8.58  .  0
  1  5.10  8.36  .  0

  ... more lines ...

12 24.15  1.17  .  0
run;
```

The following macro statement converts the Theoph_pk data set to a new data set:

```
%pkconvrt (data=Theoph_pk,out=Theoph_nlm);
```

If the conversion is successful, the following message is displayed:

```
NOTE: The data set Theoph_nlm had 144 observations and 15 variables.
The PKConvrt macro ended successfully.
```

%PKCONVRT Autocall Macro and the CMPTMODEL Statement

After you use the %PKCONVRT autocall macro to convert a data set, you can use PROC NL MIXED to fit compartment models to the data. Here the converted data set Theoph_nlm is used as an input data set:

```
proc nlmixed data=Theoph_nlm ;
  parms beta1=-3.22 beta2=0.47 beta3=-2.45
    s2b1 =0.03 cb12 =0 s2b2 =0.4 s2=0.5;
  cl = exp(beta1 + b1);
  ka = exp(beta2 + b2);
  ke = exp(beta3);
  vl = cl/ke;
  CMPTMODEL ncomps=1 admtype=oral parmtime=1 pconc=predConc
    ka=ka ke=ke scale1=vl;
  model dv ~ normal(predConc, s2);
  random b1 b2 ~ normal([0,0], [s2b1,cb12,s2b2]) subject=id;
run;
```

This program produces estimates identical to those in the earlier theophylline analysis (results not shown). It is worth noting that the DOSE= and TIME= options are not specified in this CMPTMODEL statement. The reason is because the data set that is converted by the %PKCONVRT macro already contains this information. When you use this converted data set, the CMPTMODEL statement uses this information from the data set and computes the predicted concentrations.

Examples

This section presents two pharmacokinetics examples in which the data are fitted using the CMPTMODEL statement in PROC NL MIXED. The first example uses multiple-dosage data. The second example is a combined model that includes a pharmacodynamic (PD) component.

Phenobarbital Experiment

Vonesh (2012) studied the pharmacokinetics of the seizure-prevention drug phenobarbital by analyzing data from a random experiment, in which the drug was given to 59 infants within 16 days of birth. After an initial dose of phenobarbital, each infant was given one or more intermittent sustaining doses by intravenous bolus. During this 16-day period, the serum concentrations of phenobarbital in the blood were measured randomly in each infant. Vonesh (2012) fitted the data by using the following one-compartment model,

$$C(t) = \frac{D_{t_d}}{V} \times \exp \left\{ -\frac{CL}{V}(t - t_d) \right\}$$

where D_{t_d} is the dose that is given at time t_d , $C(t)$ is the predicted concentration at time t following a dose that is administered at time t_d , V is volume, and CL is clearance. The first few observations of the original data set are shown in Figure 4.²

Figure 4 Phenobarbital Data Set

Obs	ID	time	amt	weight	apgar	dv	evid
1	1	0.0	25.0	1.4	7	.	1
2	1	2.0	.	1.4	7	17.3	0
3	1	12.5	3.5	1.4	7	.	1
4	1	24.5	3.5	1.4	7	.	1
5	1	37.0	3.5	1.4	7	.	1
6	1	48.0	3.5	1.4	7	.	1
7	1	60.5	3.5	1.4	7	.	1
8	1	72.5	3.5	1.4	7	.	1
9	1	85.3	3.5	1.4	7	.	1
10	1	96.5	3.5	1.4	7	.	1
11	1	108.5	3.5	1.4	7	.	1
12	1	112.5	.	1.4	7	31.0	0

To prepare the phenobarbital data set for analysis, you first call the %PKCONVRT autocall macro to modify the data set:

```
%pkconvrt (data=Pheno, out=Pheno_nlm);
```

Then you can use the CMPTMODEL statement to fit a one-compartment model with intravenous bolus administration:

```
proc nlmixed data = pheno_nlm method = firo;
  parms beta1= 0.01 0.5 beta2= 0.01 0.1 1
        sc1 = 0.000005 0.0005 0.05 sc2 = 0.01 0.1 sc = 0.1 3 /best = 1;
  bounds beta1 > 0, beta2 > 0;
  bounds sc1 > 0, sc2 > 0, sc > 0;
  clrn = beta1/100 + b1;
  volm = beta2 + b2;
  cmptmodel ncomps=1 admtype=ivb parmtime=2 pconcc=pred
            cl=clrn vol=volm
            scale1=volm;
  model dv ~ normal(pred,sc);
  random b1 b2 ~ normal([0,0], [sc1,0,sc2]) subject = ID;
run;
```

²Two variable names in the Vonesh (2012) data set (Dose and CONC) were renamed (AMT and DV) to meet the variable-naming requirement for the %PKCONVRT macro input data set.

In the CMPTMODEL statement, the NCOMPS=1 and ADMTYPE=IVB options require a one-compartment model with an intravenous bolus administration type. The PARMTYPE=2 option parameterizes the model in terms of clearance and volume, which require the specification of the CL= (clearance) and VOL= (volume) options. Both the clearance parameter (clrn) and the volume parameter (volm) are functions of the model parameters (beta1 and beta2) and random-effects parameters (b1 and b2). The SCALE1= option scales the predicted concentration value (pred) by volm.

The first-order method (Beal and Sheiner 1982, 1988) is applicable to the normally distributed responses and is widely used for PK applications; it is used here (METHOD=FIRO). The example uses a grid of values for each parameter as a way to identify better starting values (Vonesh 2012). The BOUNDS statement restricts the support set of some parameters: beta1 and beta2 population clearance and volume parameters that must be strictly positive, as must be the variance parameters.

Estimated parameter values are shown in Figure 5.

Figure 5 Parameter Estimates for the One-Compartment Model Fitted to Phenobarbital Data

The NLMIXED Procedure

Parameter Estimates									
Parameter	Estimate	Standard Error	DF	t Value	Pr > t	95% Confidence Limits		Gradient	Active BC
beta1	1E-8	.	57	7.40470	Lower BC
beta2	1.1325	0.08519	57	13.29	<.0001	0.9619	1.3031	-1.78979	
sc1	4.607E-7	0.000015	57	0.03	0.9750	-0.00003	0.000030	73926.3	
sc2	0.1617	0.06133	57	2.64	0.0108	0.03887	0.2845	0.88020	
sc	171.10	23.6839	57	7.22	<.0001	123.67	218.53	0.003198	

In this example, the estimates are unstable and PROC NLMIXED has a difficult time finding reasonable solutions (for example, see beta1). An alternative approach is to treat all doses as continuous and assume that the effects are cumulative (Vonesh 2012), regardless of whether intermediate concentration measurements are taken.

To make all the doses continuous, you assign EVID=1 to all observations in the data set as follows:

```
data Pheno_pkc;
  set Pheno;
  evid = 1;
  if amt = . then amt = 0;
run;

%pkconvrt (data=Pheno_pkc, out=Pheno_nlmc);
```

The modified data set Pheno_nlmc contains information that the CMPTMODEL statement needs in order to fit a model to multiple-dosage data that treats all doses as continuous. The same PROC NLMIXED program (not repeated here), but where you specify DATA=Pheno_nlmc in the PROC NLMIXED statement, leads to more reasonable estimates (Figure 6), comparable to estimates reported by Vonesh (2012).

Figure 6 Parameter Estimates for One-Compartment of Phenobarbital Data (Continuous Doses)

The NLMIXED Procedure

Parameter Estimates									
Parameter	Estimate	Standard Error	DF	t Value	Pr > t	95% Confidence Limits		Gradient	
beta1	0.5482	0.04094	57	13.39	<.0001	0.4662	0.6302	0.000148	
beta2	1.3981	0.07374	57	18.96	<.0001	1.2504	1.5457	-0.00202	
sc1	6.837E-6	1.209E-6	57	5.66	<.0001	4.416E-6	9.257E-6	-1766964	
sc2	0.2857	0.08925	57	3.20	0.0022	0.1069	0.4644	0.002769	
sc	8.0068	1.4549	57	5.50	<.0001	5.0934	10.9202	0.000059	

Vonesh (2012) fitted the same model by using a recursive representation of the multiple-dosage compartment model formula, which involves more programming effort (see example 7.2.2 in chapter 7 of Vonesh 2012) than a CMPTMODEL statement in a PROC NL MIXED call.

Argatroban Experiment

This section illustrates how to combine PK and PD models in PROC NL MIXED. There are two approaches (Davidian and Giltinan 1995): the two-stage approach (also known as the sequential approach) and the joint (or simultaneous) approach. The two-stage approach divides the problem into two steps: the first step is to fit a PK model that produces fitted (predicted) values of the PD responses; the second step is to fit a PD model to the predicted responses separately. The two-stage approach requires two separate PROC NL MIXED calls. The joint approach combines the two fittings simultaneously in one PROC NL MIXED call. This section illustrates how to fit a PK/PD model by using both approaches in PROC NL MIXED.

Argatroban is an anticoagulant that is used to treat thrombosis in patients who have heparin-induced thrombocytopenia. The kinetics of this drug has been studied in an experiment in which 37 patients were given the drug by intravenous infusion for four hours at different rates. During and after the infusion, a series of blood samples were drawn from each patient. Argatroban levels (concentrations) were measured in some of these samples, and the activated partial thromboplastin time (aPTT) was measured in the rest of the samples. Davidian and Giltinan (1995) analyzed these data by using a number of PK/PD models. The data for the first patient are shown in Figure 7.

Figure 7 Argatroban Experiment Data for First Patient

ObsNo	Subject	Rate	Time	Response	Ind
1	1	1	0	26.1	0
2	1	1	120	42.7	0
3	1	1	240	43.0	0
4	1	1	270	35.4	0
5	1	1	300	33.9	0
6	1	1	360	26.3	0
7	1	1	30	95.7	1
8	1	1	60	122.0	1
9	1	1	90	133.0	1
10	1	1	160	162.0	1
11	1	1	200	200.0	1
12	1	1	240	172.0	1
13	1	1	245	122.0	1
14	1	1	250	120.0	1
15	1	1	260	60.6	1
16	1	1	275	70.0	1
17	1	1	295	47.3	1

Both the argatroban concentrations and aPTT measurements are stored as the Response variable, distinguished by the Ind variable: Ind = 1 is an argatroban concentration (also referred to as the PK concentration), and 0 is an aPTT measurement (also referred to as the PD measurement).

Davidian and Giltinan (1995) consider a one-compartment model with intravenous infusion as a structural PK model for the argatroban concentrations.

Two-Stage Approach to PK/PD Modeling

In the first stage of the two-stage approach, you specify the response values for the PD measurements as missing and fit the PK model to the nonmissing PK responses:³

³This data set is not a PK-DataSet, so there is no need to call the %PKCONVRT macro to convert it. The CMPTMODEL statement is directly applicable because all three required variables (TIME, RATE, and duration (240 minutes)) in a one-compartment infusion model are known.

```

data Arg;
  set Argatroban;
  if ind eq 1 then conc = response;
  else conc = .;
run;

proc nlmixed data = Arg;
  parms cl=-6.0 vl=-2.0 s2b1=0.14 cb12=0.006 s2b2=0.005 s2=20.0 delta=0.5;
  cli = exp(cl + b1);
  vli = exp(vl + b2);
  durn = 240;
  amt = rate*durn;
  CMPTMODEL ncomps=1 admtype=inf parmtime=2 pconcp=pred time=time
    cl=cli vol=vli rate = rate
    dose=amt scale1=vli;
  if pred eq 0 then pred = 1e-12;
  model conc ~ normal(pred, (pred**(2*delta))*s2);
  random b1 b2 ~ normal([0,0], [s2b1,cb12,s2b2]) subject=subject;
  predict pred out=PredConc;
run;

```

The options NCOMPS=1 and ADMTYPE=INF request fitting of a one-compartment model with infusion administration. The PARMTYPE=2 option parameterizes the model in terms of clearance and volume constants. The infusion model requires either the rate of infusion or the duration of infusion. This model specifies the rate (RATE= option), which is a variable in the data set Arg. The duration of the infusion is four hours, or 240 minutes, and it is assigned to the Durn variable. The amount of drug that is administered is equal to RATE × Durn, which is specified in the DOSE= option.

The statement `if pred eq 0 then pred = 1e-12` prevents invalid likelihood calculation: when the predicted concentration variable is zero (which happens in observations that have a time of zero), the normal variance becomes zero. Adding noise to the pred variable prevents this problem.

Of the 694 observations in the data set, 475 observations with nonmissing response values (PK concentrations) are used to fit the model, and the remaining 219 PD observations are ignored (this information is displayed in the “Dimensions” output table, not shown here). The PREDICT statement computes the predicted concentration values for all observations, including those for the PD observations, and stores the predictions in the output data set PredConc. The parameter estimates from the PK model are shown in Figure 8.

Figure 8 Parameter Estimates for the PK Model in the Two-Stage Approach

Parameter Estimates								
Parameter	Estimate	Standard Error	DF	t Value	Pr > t	95% Confidence Limits		Gradient
cl	-5.4233	0.06463	35	-83.91	<.0001	-5.5546	-5.2921	0.000287
vl	-1.8691	0.03843	35	-48.63	<.0001	-1.9471	-1.7910	0.000214
s2b1	0.1498	0.03595	35	4.17	0.0002	0.07676	0.2227	-0.00038
cb12	0.01371	0.01208	35	1.13	0.2643	-0.01082	0.03824	-0.00442
s2b2	0.01126	0.007517	35	1.50	0.1431	-0.00400	0.02652	0.000108
s2	124.40	83.8452	35	1.48	0.1468	-45.8125	294.62	2.398E-6
delta	0.3413	0.05627	35	6.07	<.0001	0.2270	0.4555	0.004138

In the second stage of the two-stage approach, you use the predicted responses of the aPTT measurements, the PD part of the data. There are a number of PD models that you can choose from to study the relationship between the predicted concentrations and the PD measurements. A popular choice is the E_{max} model (Davidian and Giltinan 1995),

$$y_{ij} = E_{0i} + \frac{E_{maxi} - E_{0i}}{1 + EC_{50i}/C_p(t_{ij})} + e_{ij}$$

where $C_p(t_{ij})$ is the predicted concentration.

The following PROC NL MIXED program fits the E_{\max} model by using the PD portion of the data:

```
proc nlmixed data = PredConc (where=(ind=0));
  parms E0= 25 Emax= 70 EC50 = 400
        s2b1=6 s2b2=25 s2b3= 100 cb12=1 cb13=1 cb23=1 s2=20.0 delta=0.5;

  E0i   = E0   + b3;
  Emaxi = Emax + b4;
  EC50i = EC50 + b5;
  if pred eq 0 then pd_mean = E0i;
  else pd_mean = E0i + (Emaxi - E0i)/(1+(EC50i/pred));

  model response ~ normal(pd_mean, (pd_mean**(2*delta))*s2);
  random b3 b4 b5 ~ normal([0,0,0], [s2b1,
                                     cb12, s2b2,
                                     cb13, cb23, s2b3]) subject=subject;

run;
```

The WHERE suboption in the DATA= option selects observations in which IND=0 (PD measurements) to fit in the PROC NL MIXED run. In observations where the predicted concentrations (PK) are zero, the mean of the response reduces to the baseline value E_0 . The parameter estimates are shown in Figure 9.

Figure 9 Parameter Estimates for PD Model in the Two-Stage Approach
The NL MIXED Procedure

Parameter	Parameter Estimates					95% Confidence Limits		Gradient
	Estimate	Standard Error	DF	t Value	Pr > t			
E0	28.7236	0.6076	34	47.27	<.0001	27.4888	29.9584	-0.00080
Emax	75.1921	4.5486	34	16.53	<.0001	65.9482	84.4360	-0.00202
EC50	419.58	84.1595	34	4.99	<.0001	248.55	590.61	-0.03661
s2b1	9.4465	4.1554	34	2.27	0.0294	1.0017	17.8913	-0.00012
s2b2	113.03	62.9064	34	1.80	0.0813	-14.8096	240.87	0.010509
s2b3	100.11	50.3754	34	1.99	0.0550	-2.2663	202.48	0.000023
cb12	28.6852	16.5223	34	1.74	0.0916	-4.8921	62.2625	0.000573
cb13	-7.2150	238.57	34	-0.03	0.9761	-492.04	477.61	0.010703
cb23	-4.4550	646.75	34	-0.01	0.9945	-1318.81	1309.90	-0.00169
s2	0.01141	0.01940	34	0.59	0.5603	-0.02802	0.05085	-0.37673
delta	0.9352	0.2280	34	4.10	0.0002	0.4719	1.3986	-0.03230

In general, the two-stage or sequential modeling approach is faster than the joint modeling approach, but it tends to produce larger errors in PD parameter estimates. However, in certain suitable designs (see Zhang, Beal, and Sheiner 2003), the two-stage modeling approach can achieve the same accuracy as the joint modeling approach.

Joint Approach to PK/PD Modeling

You can fit the PK model and the PD model in a single PROC NL MIXED call. The program is slightly more complicated, but it does not require much more than combining the two separate procedure calls that are used in the two-stage approach:

```
proc nlmixed data = Argatroban;
  parms cl=-6.0 v1=-2.0 E0= 25 Emax= 70 EC50 = 400
        s2b1=0.14 cb12=0.006 s2b2=0.005
        s2b3=9.4 s2b4=113 s2b5=100 cb34=28 cb35=-7.2 cb45=-4.4
        s2=20.0 delta=0.5;
  cli = exp(cl + b1);
  vli = exp(v1 + b2);
```

```

durn = 240;
dose = rate*durn;
CMPTMODEL ncomps=1 admtype=inf parmtype=2 pconcp=pk_mean time=time
  cl=cli vol=vli rate = rate
  dose=dose scale1=vli;

E0i = E0 + b3;
Emaxi = Emax + b4;
EC50i = EC50 + b5;
if pk_mean eq 0 then pd_mean = E0i;
else pd_mean = E0i + (Emaxi - E0i)/(1+(EC50i/pk_mean));

if(ind eq 1) then resp_mean = pk_mean;
if(ind eq 0) then resp_mean = pd_mean;

model response ~ normal(resp_mean, (resp_mean**(2*delta))*s2);
random b1 b2 b3 b4 b5 ~ normal([0,0,0,0,0],[s2b1,
                                cb12,s2b2,
                                0, 0, s2b3,
                                0, 0, cb34, s2b4,
                                0, 0, cb35, cb45, s2b5]) subject=subject;

run;

```

The program includes two sets of model parameters and five-dimensional random effects (two for the PK model and three for the PD model). The IF statement sets the response mean according to either the predicted PK mean or the predicted PD mean.

The parameter estimates are shown in Figure 10.

Figure 10 Parameter Estimates for the PK/PD Model in the Joint Approach
The NLMIXED Procedure

Parameter	Parameter Estimates							
	Estimate	Standard Error	DF	t Value	Pr > t	95% Confidence Limits		Gradient
cl	-5.4488	0.06789	32	-80.26	<.0001	-5.5871	-5.3105	-0.44025
vl	-1.6752	0.03186	32	-52.57	<.0001	-1.7401	-1.6103	-0.74288
E0	27.7414	0.6605	32	42.00	<.0001	26.3959	29.0868	0.71257
Emax	71.1146	3.2128	32	22.13	<.0001	64.5703	77.6590	-0.37873
EC50	400.12	43.5750	32	9.18	<.0001	311.36	488.88	-0.02001
s2b1	0.1665	0.03925	32	4.24	0.0002	0.08650	0.2464	-1.12496
cb12	0.04385	0.01516	32	2.89	0.0068	0.01297	0.07473	0.76753
s2b2	0.02682	0.008352	32	3.21	0.0030	0.009811	0.04384	-1.20904
s2b3	8.0010	4.7027	32	1.70	0.0986	-1.5781	17.5801	0.55501
s2b4	112.78	70.6242	32	1.60	0.1201	-31.0777	256.64	0.029990
s2b5	99.9992	45.2913	32	2.21	0.0345	7.7438	192.25	0.000116
cb34	28.6197	18.7122	32	1.53	0.1360	-9.4958	66.7352	-0.19546
cb35	-7.2379	263.70	32	-0.03	0.9783	-544.38	529.91	0.013052
cb45	-4.3754	832.14	32	-0.01	0.9958	-1699.39	1690.64	-0.00371
s2	0.003086	0.001222	32	2.53	0.0167	0.000598	0.005574	4.32154
delta	1.2256	0.03592	32	34.12	<.0001	1.1524	1.2988	0.018653

The parameter estimates are similar to the results of the two-stage approach (Figure 8 and Figure 9). However, in this analysis, both approaches are prone to computational instability and convergence issues that can lead to a nonpositive definite Hessian matrix. Because of a potential postconvergence issue of unreliability in the final parameter estimates, Davidian and Giltinan (1995) were not able to conclude that the two model approaches would yield the same results in this analysis. Generally speaking, good initial values and suitable optimization techniques are important to getting clean convergence in this class of problems.

Bayesian Analysis

PROC MCMC is similar to PROC NL MIXED: both procedures can be used to fit nonlinear random-effects models, and both offer flexibility in supporting SAS[®] programming statements. PROC MCMC provides full Bayesian solutions, relying mostly on Markov chain Monte Carlo sampling algorithms to estimate the posterior distributions of the parameters. PROC MCMC supports the CMPTMODEL statement, and you can use the procedure to fit Bayesian PK models.

However, there is one important difference between these procedures when it comes to using the %PKCONVRT macro to convert a PK-DataSet. To use the data set in PROC MCMC, you must first remove all dosing observations. You do this by specifying the DOSEDROP=TRUE option in the macro statement. The data set from the theophylline example in the section “%PKCONVRT Autocall Macro” on page 10 is used here to illustrate:

```
%pkconvrt (data=Theoph_pk, out=Theoph_mcmc, dosedrop=TRUE);
```

This option removes all dosing event observations (EVID=1), which typically have missing values in the DV variable (concentration measurements). But missing DV values in PROC NL MIXED do not pose a problem, because the procedure ignores observations that have missing response values. However, by default, PROC MCMC treats missing response values as parameters and incorporates the sampling of these missing values in a Markov chain simulation. As a result, dosage observations end up contributing to the estimation of the model parameters; this violates modeling assumptions and leads to different inferences. To avoid this pitfall, you use the DOSEDROP=TRUE option to clean up the output data set by retaining only measurement events in the data set.

Now you can use PROC MCMC to fit a PK model:

```
proc mcmc data=Theoph_mcmc seed=170730 ntu=1000 nmc=50000 outpost=tOut;
  array b[2];
  array mu[2] (0 0);
  array cov[2,2];
  array S[2,2] (1 0 0 1);

  parms beta1=-3.22 beta2=0.47 beta3=-2.45;
  parms cov {0.03 0 0 0.4} s2y;
  prior beta: ~ n(0, sd=10000);
  prior cov ~ iwish(2, S);
  prior s2y ~ igamma(shape=3, scale=2);
  random b ~ mvn(mu, cov) subject=id;

  cl = exp(beta1 + b1);
  ka = exp(beta2 + b2);
  ke = exp(beta3);
  vl = cl/ke;
  CMPTMODEL ncomps=1 admtype=oral parmtime=1 pconc=predConc
    ka=ka ke=ke scale1=vl;
  model dv ~ normal(predConc, var=s2y);
run;
```

This program looks very similar to the PROC NL MIXED program that fits the same model: it has identical specifications in the CMPTMODEL statement and its options. Because of its Bayesian nature, PROC MCMC requires input of prior distributions to all model parameters.

Predictions with Existing Subjects

After you obtain parameter estimates, you can compute predicted concentration values for each subject (predictive inferences). This is a way to check how well the model fits the data or to make predictions about new covariates (at a new time, for example).

If you denote y as observed data, \tilde{y} as the new response, θ as the fixed-effects parameters (which can include the regression parameter vector β , the random-effects covariance Σ , and the sample variance parameter σ^2), and γ as the random effects, then the Bayesian predictive distribution is $p(\tilde{y}|y) = \int p(\tilde{y}|\theta, \gamma)\pi(\theta, \gamma|y)d(\theta, \gamma)$. The PREDDIST statement in PROC MCMC generates samples from the $p(\tilde{y}|y)$. You can add the following statement to the PROC MCMC program:

```
preddist outpred=Theoph_pred;
```


This statement produces distributional samples of the response (DV) variable for every observation in the input data set.

The PREDDIST statement in PROC MCMC is different from the PREDICT statement in PROC NL MIXED. The PREDICT statement in PROC NL MIXED computes a user-input expression across observations in the input data set, using estimates as plug-ins ($\hat{\theta}, \hat{y}$). In the theophylline example, the following PREDICT statement in PROC NL MIXED computes the expected value of the concentrated response ($E(\tilde{y}|\hat{\theta}, \hat{y})$):

```
predict predConc out=PredConc;
```

In addition, the PREDICT statement computes the standard errors on the basis of these estimates. The statement does not account for the uncertainties that are associated with the parameter estimates (for example, the random effects). On the other hand, the PREDDIST statement in PROC MCMC produces samples from the marginal predictive distribution of the response variable, with all parameter uncertainties integrated out (using Monte Carlo methods). You can see the two different types of predictions in Figure 11 for the first four subjects in the Theophylline data set.

Figure 11 Two Types of Predictions for the First Four Subjects in the Theophylline Example



The solid lines represent the observed response (DV). The predicted values from the two procedures are very close to each other and almost indistinguishable (the short-dashed lines are Bayesian estimates; the long-dashed lines are PROC NL MIXED estimates). But the predictive intervals are different: the Bayesian highest posterior density (HPD) intervals are substantially wider than the frequentist confidence intervals. Again, this is because the Bayesian intervals account for the uncertainties in the random effects, but the plug-in methods do not.

You can combine PROC NL MIXED and PROC MCMC to make frequentist-based prediction of the response variables while accounting for uncertainties of the random effects:

$$p(\tilde{y}|y, \theta = \hat{\theta}) = \int p(\tilde{y}|\hat{\theta}, \gamma)p(\gamma|y)d\gamma$$

You take estimates from PROC NL MIXED, plug them into a PROC MCMC program (which becomes a parameter-less model, except for the random effects), and use PROC MCMC to sample predictive response values.

The following program takes parameter estimates $(\hat{\beta}, \hat{\Sigma}, \hat{\sigma}_y^2)$ from [Figure 3](#), treats them as known constants in a model, and uses the PREDDIST statement to sample predictive samples of DV:

```
proc mcmc data=Theoph_mcmc seed=170730 ntu=1000 nmc=50000 outpost=tOut;
  array b[2];
  array mu[2] (0 0);
  array cov[2,2];

  begincnst;
  beta1 = -3.2268; beta2 = 0.4806; beta3 = -2.4592;
  cov[1,1] = 0.02803; cov[1,2] = -0.00127;
  cov[2,1] = cov[1,2]; cov[2,2] = 0.4331;
  s2y = 0.5016;
  endcnst;

  random b ~ mvn(mu, cov) subject=id;
  c1 = exp(beta1 + b1);
  ka = exp(beta2 + b2);
  ke = exp(beta3);
  v1 = c1/ke;
  CMPTMODEL ncomps=1 admtype=oral parmtime=1 pconc=predConc
            ka=ka ke=ke scale1=v1;
  model dv ~ normal(predConc, var=s2y);
  preddist outpred=Theoph_predF;
run;
```

There are no PARM or PRIOR statements in the PROC MCMC program because the model parameters are treated as constants and given assigned values. The only random variables are the random effects, which you want to integrate out in the prediction. The prediction intervals (not shown here) are very similar to the Bayesian prediction intervals in [Figure 11](#).

Predictions with New Subjects

You can also make inference based on the predictive distribution of a new subject; that is, you can generate predictive samples from hypothetical patients who are not in the original study design to see how the predictive responses behave over time. The PREDDIST statement produces samples $(p(\tilde{y}|y))$ for a new subject) on the basis of input data sets that represent covariate information from new subjects. In the following simulated data set, six new subjects (represented by ID values that are not in the Theophylline input data set) are given different dosage amounts:

```
data AMT;
  retain ID 100;
  do AMT = 3.5 to 6 by 0.5;
    ID+1; output;
  end;
run;

data NewDATA;
  set AMT;
  EVID = 1;
  DV = .;
  do Time = 0 to 24 by 0.5;
    if Time > 0 then EVID = 0;
    output;
  end;
run;
```


You add this prediction data set as a covariate in the PREDDIST statement in PROC MCMC:

```
preddist covariates=NewDATA_mcmc outpred=Theoph_predN;
```

The OUTPRED= data set stores posterior predictive samples for each new patient at every time grid.

You can use a similar approach to combine PROC NL MIXED estimates in a PROC MCMC call to draw samples from the predictive distribution of a new patient, conditioned on estimated fixed-effects parameters: $p(\tilde{y}|y, \hat{\theta})$. This is easy to do: you repeat the previous PROC MCMC program, adding the NewDATA_mcmc data set as the covariates data set in the PREDDIST statement. The results are shown in Figure 12.

Figure 12 Predicted Profiles from $p(\tilde{y}|y, \hat{\theta})$ for New Patients Given Different Dosages



Conclusion

In clinical trials, determining the pharmacokinetics of a new drug by using structural PK models is a major step in establishing the dose regimen for the drug. Compartment models are the basic building blocks of these PK models, and one-, two-, and three-compartment models are widely used in relevant stages of drug development. The new CMPTMODEL statement, which is supported in PROC NL MIXED and PROC MCMC in SAS/STAT 14.3, provides you with convenience and flexibility in fitting these models, in frequentist as well as Bayesian approaches. The %PKCONVRT autocall macro provides a convenient way to modify industry-standard PK data for easy usage in SAS procedures. And you can use the array of functionality in PROC NL MIXED and PROC MCMC to fit PK/PD models by using a two-stage or joint modeling approach, as well as to perform postfitting and make predictions.

REFERENCES

- Abuhelwa, A. Y., Foster, D. J., and Upton, R. N. (2015). "ADVAN-Style Analytical Solutions for Common Pharmacokinetic Models." *Journal of Pharmacological and Toxicological Methods* 73:42–48.
- Beal, S. L., and Sheiner, L. B. (1982). "Estimating Population Kinetics." *CRC Critical Reviews in Biomedical Engineering* 8:195–222.

- Beal, S. L., and Sheiner, L. B. (1988). "Heteroscedastic Nonlinear Regression." *Technometrics* 30:327–338.
- Bonate, P. L. (2011). *Pharmacokinetic-Pharmacodynamic Modeling and Simulation*. 2nd ed. New York: Springer.
- Davidian, M., and Giltinan, D. M. (1995). *Nonlinear Models for Repeated Measurement Data*. New York: Chapman & Hall.
- Fisher, D., and Shafer, S. (2007). "Pharmacokinetic and Pharmacodynamic Analysis with NONMEM: Basic Concepts." Course materials for Fisher/Shafer NONMEM Workshop, March 7–11, Ghent, Belgium.
- Gabrielsson, J., and Weiner, D. (2006). *Pharmacokinetic and Pharmacodynamic Data Analysis: Concepts and Applications*. 4th ed. Stockholm: Swedish Pharmaceutical Press.
- Mould, D. R., and Upton, R. N. (2012). "Basic Concepts in Population Modeling, Simulation, and Model-Based Drug Development." *CPT: Pharmacometrics and Systems Pharmacology* 1:e6.
- Owen, J. S., and Fiedler-Kelly, J. (2014). *Introduction to Population Pharmacokinetic/Pharmacodynamic Analysis with Nonlinear Mixed Effects Models*. Hoboken, NJ: John Wiley & Sons.
- Pinheiro, J. C., and Bates, D. M. (1995). "Approximations to the Log-Likelihood Function in the Nonlinear Mixed-Effects Model." *Journal of Computational and Graphical Statistics* 4:12–35.
- Vonesh, E. F. (2012). *Generalized Linear and Nonlinear Models for Correlated Data: Theory and Applications Using SAS*. Cary, NC: SAS Institute Inc.
- Zhang, L., Beal, S. L., and Sheiner, L. B. (2003). "Simultaneous vs. Sequential Analysis for Population PK/PD Data II: Robustness of Methods." *Journal of Pharmacokinetics and Pharmacodynamics* 30:405–416.

Appendix

Solutions to the One-, Two-, and Three-Compartment Model Differential Equations

As mentioned in the section "Compartment Models and Differential Equations" on page 3, the amount of drug in each compartment is expressed by using a set of differential equations for any compartment model. The differential equations for one-, two-, and three-compartment models are solved analytically in Abuhelwa, Foster, and Upton (2015) and Fisher and Shafer (2007) for both intravenous and extravascular drug administration types. In this section, these results are presented again by using the following notation:

Amount of drug ($A_i(t)$): the amount of drug present in the i th compartment at time t , for $i = 1, 2, 3$

Concentration ($C_i(t)$): the plasma drug concentration in compartment i , which is calculated as the amount of drug per volume. C_1 , C_2 , and C_3 are the concentrations in compartments 1, 2, and 3, respectively.

Clearance (CL_i): the volume of the plasma cleared of the drug per unit time. CL_1 , CL_2 , and CL_3 are the clearance constants for compartments 1, 2, and 3, respectively.

Volume (VOL_i): the volume of distribution, where the drug is uniformly distributed. VOL_1 , VOL_2 , and VOL_3 are the volumes of compartments 1, 2, and 3, respectively.

Elimination rate (K_{i0}): the elimination rate constant, which characterizes the excretion process for the i th compartment, for $i = 1, 2, 3$. The central compartment elimination rate constant is also denoted as K_e .

Absorption rate (K_a): the absorption rate constant, which characterizes the rate of absorption of a drug

Transfer rate (K_{ij}): the transfer rate constant from compartment i to compartment j . K_{12} , K_{21} , K_{13} , and K_{31} are used for two- and three-compartment models.

Initial dose (D_i): the initial dosage of the drug given in the i th compartment, for $i = 1, 2, 3$

In addition, for the oral compartment models, $A_0(t)$ and D_0 are denoted as the amount of drug at time t and the initial dose administered in the depot compartment, respectively.

One-Compartment Model

1. Intravenous bolus dose:

$$A_1(t) = D_1 e^{-k_{10}t}$$

2. Intravenous dose with a constant rate of infusion:

$$A_1(t) = \begin{cases} \frac{\text{Rate}}{k_{10}} (1 - e^{-k_{10}t}), & \text{if } t \leq \text{Durn} \\ \frac{\text{Rate}}{k_{10}} (1 - e^{-k_{10} \times \text{Durn}}) e^{-k_{10}(t - \text{Durn})}, & \text{otherwise} \end{cases}$$

3. First-order absorption with oral dose:

$$A_0(t) = D_0 e^{-k_a t}$$

$$A_1(t) = \frac{D_0 k_a}{k_a - k_{10}} [e^{-k_{10}t} - e^{-k_a t}] + D_1 e^{-k_{10}t}$$

Two-Compartment Model

The following quantities are required for the analytical solutions of the two-compartment model:

$$E_1 = k_{10} + k_{12}$$

$$E_2 = k_{20} + k_{21}$$

$$\lambda_1 = \frac{1}{2} \left[(E_1 + E_2) + \sqrt{(E_1 + E_2)^2 - 4(E_1 E_2 - k_{12} k_{21})} \right]$$

$$\lambda_2 = \frac{1}{2} \left[(E_1 + E_2) - \sqrt{(E_1 + E_2)^2 - 4(E_1 E_2 - k_{12} k_{21})} \right]$$

Using these quantities, the predicted amount of the drug in compartments 1 and 2 is calculated as follows:

1. Intravenous bolus dose:

$$A_1(t) = \frac{[\{D_1 E_2 + D_2 k_{21}\} - D_1 \lambda_1] e^{-\lambda_1 t} - [\{D_1 E_2 + D_2 k_{21}\} - D_1 \lambda_2] e^{-\lambda_2 t}}{\lambda_2 - \lambda_1}$$

$$A_2(t) = \frac{[\{D_2 E_1 + D_1 k_{12}\} - D_2 \lambda_1] e^{-\lambda_1 t} - [\{D_2 E_1 + D_1 k_{12}\} - D_2 \lambda_2] e^{-\lambda_2 t}}{\lambda_2 - \lambda_1}$$

2. Intravenous dose with a constant rate of infusion:

$$A_1(t) = \begin{cases} \text{Rate} \times \left[\frac{k_{21} - \lambda_1}{\lambda_1 (\lambda_2 - \lambda_1)} (1 - e^{-\lambda_1 t}) + \frac{k_{21} - \lambda_2}{\lambda_2 (\lambda_1 - \lambda_2)} (1 - e^{-\lambda_2 t}) \right], & \text{if } t \leq \text{Durn} \\ \text{Rate} \times \left[\frac{k_{21} - \lambda_1}{\lambda_1 (\lambda_2 - \lambda_1)} (1 - e^{-\lambda_1 \times \text{Durn}}) e^{-\lambda_1 (t - \text{Durn})} \right. \\ \quad \left. + \frac{k_{21} - \lambda_2}{\lambda_2 (\lambda_1 - \lambda_2)} (1 - e^{-\lambda_2 \times \text{Durn}}) e^{-\lambda_2 (t - \text{Durn})} \right], & \text{otherwise} \end{cases}$$

3. First-order absorption with oral dose:

$$A_0(t) = D_0 e^{-k_a t}$$

$$A_1(t) = D_0 k_a \left[\frac{E_2 - k_a}{(\lambda_1 - k_a)(\lambda_2 - k_a)} e^{-k_a t} + \frac{E_2 - \lambda_1}{(\lambda_2 - \lambda_1)(k_a - \lambda_1)} e^{-\lambda_1 t} \right. \\ \left. + \frac{E_2 - \lambda_2}{(\lambda_1 - \lambda_2)(k_a - \lambda_2)} e^{-\lambda_2 t} \right] \\ + \frac{[\{D_1 E_2 + D_2 k_{21}\} - D_1 \lambda_1] e^{-\lambda_1 t} - [\{D_1 E_2 + D_2 k_{21}\} - D_1 \lambda_2] e^{-\lambda_2 t}}{\lambda_2 - \lambda_1}$$

$$A_2(t) = D_0 k_a k_{12} \left[\frac{e^{-k_a t}}{(\lambda_1 - k_a)(\lambda_2 - k_a)} + \frac{e^{-\lambda_1 t}}{(\lambda_2 - \lambda_1)(k_a - \lambda_2)} + \frac{e^{-\lambda_2 t}}{(\lambda_1 - \lambda_2)(k_a - \lambda_2)} \right] \\ + \frac{[\{D_2 E_1 + D_1 k_{12}\} - D_2 \lambda_1] e^{-\lambda_1 t} - [\{D_2 E_1 + D_1 k_{12}\} - D_2 \lambda_2] e^{-\lambda_2 t}}{\lambda_2 - \lambda_1}$$

Three-Compartment Model

The following quantities are required for the analytical solutions of the three-compartment model:

$$E_1 = k_{10} + k_{12} + k_{13}$$

$$E_2 = k_{20} + k_{21}$$

$$E_3 = k_{30} + k_{31}$$

$$a = E_1 + E_2 + E_3$$

$$b = E_1 E_2 + E_3(E_1 + E_2) - k_{12}k_{21} - k_{13}k_{31}$$

$$c = E_1 E_2 E_3 - E_3 k_{12} k_{21} - E_2 k_{13} k_{31}$$

$$m = \frac{(3b - a^2)}{3}$$

$$n = \frac{(2a^3 - 9ab + 27c)}{27}$$

$$Q = \frac{n^2}{4} + \frac{m^3}{27}$$

$$\alpha = \sqrt{-Q}$$

$$\beta = -\frac{n}{2}$$

$$\gamma = \sqrt{\beta^2 + \alpha^2}$$

$$\theta = \text{atan2}(\alpha, \beta)$$

$$B = D_2 k_{21} + D_3 k_{31}$$

$$C = D_2 k_{21} E_3 + D_3 k_{31} E_2$$

$$I = D_1 k_{12} E_3 - D_2 k_{13} k_{31} + D_3 k_{31} k_{12}$$

$$J = D_1 k_{13} E_2 - D_3 k_{12} k_{21} + D_2 k_{13} k_{21}$$

$$\lambda_1 = \frac{a}{3} + \gamma^{1/3} \left(\cos\left(\frac{\theta}{3}\right) + \sqrt{3} \sin\left(\frac{\theta}{3}\right) \right)$$

$$\lambda_2 = \frac{a}{3} + \gamma^{1/3} \left(\cos\left(\frac{\theta}{3}\right) - \sqrt{3} \sin\left(\frac{\theta}{3}\right) \right)$$

$$\lambda_3 = \frac{a}{3} - \left(2\gamma^{1/3} \cos\left(\frac{\theta}{3}\right) \right)$$

Using these quantities, the predicted amount of the drug in compartments 1, 2, and 3 is calculated as follows:

1. Intravenous bolus dose:

$$\begin{aligned} A_1(t) = & D_1 \left[\frac{(E_2 - \lambda_1)(E_3 - \lambda_1)}{(\lambda_2 - \lambda_1)(\lambda_3 - \lambda_1)} e^{-\lambda_1 t} + \frac{(E_2 - \lambda_2)(E_3 - \lambda_2)}{(\lambda_1 - \lambda_2)(\lambda_3 - \lambda_2)} e^{-\lambda_2 t} \right. \\ & \left. + \frac{(E_2 - \lambda_3)(E_3 - \lambda_3)}{(\lambda_1 - \lambda_3)(\lambda_2 - \lambda_3)} e^{-\lambda_3 t} \right] + \left[\frac{(C - B\lambda_1)}{(\lambda_1 - \lambda_2)(\lambda_1 - \lambda_3)} e^{-\lambda_1 t} \right. \\ & \left. + \frac{(B\lambda_2 - C)}{(\lambda_1 - \lambda_2)(\lambda_2 - \lambda_3)} e^{-\lambda_2 t} + \frac{(B\lambda_3 - C)}{(\lambda_1 - \lambda_3)(\lambda_3 - \lambda_2)} e^{-\lambda_3 t} \right] \end{aligned}$$

$$\begin{aligned} A_2(t) = & D_2 \left[\frac{(E_1 - \lambda_1)(E_3 - \lambda_1)}{(\lambda_2 - \lambda_1)(\lambda_3 - \lambda_1)} e^{-\lambda_1 t} + \frac{(E_1 - \lambda_2)(E_3 - \lambda_2)}{(\lambda_1 - \lambda_2)(\lambda_3 - \lambda_2)} e^{-\lambda_2 t} \right. \\ & \left. + \frac{(E_1 - \lambda_3)(E_3 - \lambda_3)}{(\lambda_1 - \lambda_3)(\lambda_2 - \lambda_3)} e^{-\lambda_3 t} \right] + \left[\frac{(I - D_1 k_{12} \lambda_1)}{(\lambda_1 - \lambda_2)(\lambda_1 - \lambda_3)} e^{-\lambda_1 t} \right. \\ & \left. + \frac{(D_1 k_{12} \lambda_2 - I)}{(\lambda_1 - \lambda_2)(\lambda_2 - \lambda_3)} e^{-\lambda_2 t} + \frac{(D_1 k_{12} \lambda_3 - I)}{(\lambda_1 - \lambda_3)(\lambda_3 - \lambda_2)} e^{-\lambda_3 t} \right] \end{aligned}$$

$$\begin{aligned} A_3(t) = & D_3 \left[\frac{(E_1 - \lambda_1)(E_2 - \lambda_1)}{(\lambda_2 - \lambda_1)(\lambda_3 - \lambda_1)} e^{-\lambda_1 t} + \frac{(E_1 - \lambda_2)(E_2 - \lambda_2)}{(\lambda_1 - \lambda_2)(\lambda_3 - \lambda_2)} e^{-\lambda_2 t} \right. \\ & \left. + \frac{(E_1 - \lambda_3)(E_2 - \lambda_3)}{(\lambda_1 - \lambda_3)(\lambda_2 - \lambda_3)} e^{-\lambda_3 t} \right] + \left[\frac{(J - D_1 k_{13} \lambda_1)}{(\lambda_1 - \lambda_2)(\lambda_1 - \lambda_3)} e^{-\lambda_1 t} \right. \\ & \left. + \frac{(D_1 k_{13} \lambda_2 - J)}{(\lambda_1 - \lambda_2)(\lambda_2 - \lambda_3)} e^{-\lambda_2 t} + \frac{(D_1 k_{13} \lambda_3 - J)}{(\lambda_1 - \lambda_3)(\lambda_3 - \lambda_2)} e^{-\lambda_3 t} \right] \end{aligned}$$

2. Intravenous dose with a constant rate of infusion:

$$A_1(t) = \begin{cases} \text{Rate} \times \left[\frac{(k_{21}-\lambda_1)(k_{31}-\lambda_1)}{\lambda_1(\lambda_1-\lambda_2)(\lambda_1-\lambda_3)} (1 - e^{-\lambda_1 t}) \right. \\ \quad \left. + \frac{(k_{21}-\lambda_2)(k_{31}-\lambda_2)}{\lambda_2(\lambda_2-\lambda_1)(\lambda_2-\lambda_3)} (1 - e^{-\lambda_2 t}) \right. \\ \quad \left. + \frac{(k_{21}-\lambda_3)(k_{31}-\lambda_3)}{\lambda_3(\lambda_3-\lambda_2)(\lambda_3-\lambda_1)} (1 - e^{-\lambda_3 t}) \right], & \text{if } t \leq \text{Durn} \\ \text{Rate} \times \left[\frac{(k_{21}-\lambda_1)(k_{31}-\lambda_1)}{\lambda_1(\lambda_1-\lambda_2)(\lambda_1-\lambda_3)} (1 - e^{-\lambda_1 \times \text{Durn}}) e^{-\lambda_1(t-\text{Durn})} \right. \\ \quad \left. + \frac{(k_{21}-\lambda_2)(k_{31}-\lambda_2)}{\lambda_2(\lambda_2-\lambda_1)(\lambda_2-\lambda_3)} (1 - e^{-\lambda_2 \times \text{Durn}}) e^{-\lambda_2(t-\text{Durn})} \right. \\ \quad \left. + \frac{(k_{21}-\lambda_3)(k_{31}-\lambda_3)}{\lambda_3(\lambda_3-\lambda_2)(\lambda_3-\lambda_1)} (1 - e^{-\lambda_3 \times \text{Durn}}) e^{-\lambda_3(t-\text{Durn})} \right], & \text{otherwise} \end{cases}$$

3. First-order absorption with oral dose:

$$A_0(t) = D_0 e^{-k_a t}$$

$$A_1(t) = D_0 k_a \left[\frac{(E_2 - k_a)(E_3 - k_a)}{(\lambda_1 - k_a)(\lambda_2 - k_a)(\lambda_3 - k_a)} e^{-k_a t} + \frac{(E_2 - \lambda_1)(E_3 - \lambda_1)}{(k_a - \lambda_1)(\lambda_2 - \lambda_1)(\lambda_3 - \lambda_1)} e^{-\lambda_1 t} \right. \\ \left. + \frac{(E_2 - \lambda_2)(E_3 - \lambda_2)}{(k_a - \lambda_2)(\lambda_1 - \lambda_2)(\lambda_3 - \lambda_2)} e^{-\lambda_2 t} + \frac{(E_2 - \lambda_3)(E_3 - \lambda_3)}{(k_a - \lambda_3)(\lambda_1 - \lambda_3)(\lambda_2 - \lambda_3)} e^{-\lambda_3 t} \right] \\ + D_1 \left[\frac{(E_2 - \lambda_1)(E_3 - \lambda_1)}{(\lambda_2 - \lambda_1)(\lambda_3 - \lambda_1)} e^{-\lambda_1 t} + \frac{(E_2 - \lambda_2)(E_3 - \lambda_2)}{(\lambda_1 - \lambda_2)(\lambda_3 - \lambda_2)} e^{-\lambda_2 t} \right. \\ \left. + \frac{(E_2 - \lambda_3)(E_3 - \lambda_3)}{(\lambda_1 - \lambda_3)(\lambda_2 - \lambda_3)} e^{-\lambda_3 t} \right] + \left[\frac{(C - B\lambda_1)}{(\lambda_1 - \lambda_2)(\lambda_1 - \lambda_3)} e^{-\lambda_1 t} \right. \\ \left. + \frac{(B\lambda_2 - C)}{(\lambda_1 - \lambda_2)(\lambda_2 - \lambda_3)} e^{-\lambda_2 t} + \frac{(B\lambda_3 - C)}{(\lambda_1 - \lambda_3)(\lambda_3 - \lambda_2)} e^{-\lambda_3 t} \right]$$

$$A_2(t) = D_0 k_a k_{12} \left[\frac{(E_3 - k_a)}{(\lambda_1 - k_a)(\lambda_2 - k_a)(\lambda_3 - k_a)} e^{-k_a t} \right. \\ \left. + \frac{(E_3 - \lambda_1)}{(k_a - \lambda_1)(\lambda_2 - \lambda_1)(\lambda_3 - \lambda_1)} e^{-\lambda_1 t} + \frac{(E_3 - \lambda_2)}{(k_a - \lambda_2)(\lambda_1 - \lambda_2)(\lambda_3 - \lambda_2)} e^{-\lambda_2 t} \right. \\ \left. + \frac{(E_3 - \lambda_3)}{(k_a - \lambda_3)(\lambda_1 - \lambda_3)(\lambda_2 - \lambda_3)} e^{-\lambda_3 t} \right] \\ + D_2 \left[\frac{(E_1 - \lambda_1)(E_3 - \lambda_1)}{(\lambda_2 - \lambda_1)(\lambda_3 - \lambda_1)} e^{-\lambda_1 t} + \frac{(E_1 - \lambda_2)(E_3 - \lambda_2)}{(\lambda_1 - \lambda_2)(\lambda_3 - \lambda_2)} e^{-\lambda_2 t} \right. \\ \left. + \frac{(E_1 - \lambda_3)(E_3 - \lambda_3)}{(\lambda_1 - \lambda_3)(\lambda_2 - \lambda_3)} e^{-\lambda_3 t} \right] + \left[\frac{(I - D_1 k_{12} \lambda_1)}{(\lambda_1 - \lambda_2)(\lambda_1 - \lambda_3)} e^{-\lambda_1 t} \right. \\ \left. + \frac{(D_1 k_{12} \lambda_2 - I)}{(\lambda_1 - \lambda_2)(\lambda_2 - \lambda_3)} e^{-\lambda_2 t} + \frac{(D_1 k_{12} \lambda_3 - I)}{(\lambda_1 - \lambda_3)(\lambda_3 - \lambda_2)} e^{-\lambda_3 t} \right]$$

$$A_3(t) = D_0 k_a k_{13} \left[\frac{(E_2 - k_a)}{(\lambda_1 - k_a)(\lambda_2 - k_a)(\lambda_3 - k_a)} e^{-k_a t} \right. \\ \left. + \frac{(E_2 - \lambda_1)}{(k_a - \lambda_1)(\lambda_2 - \lambda_1)(\lambda_3 - \lambda_1)} e^{-\lambda_1 t} + \frac{(E_2 - \lambda_2)}{(k_a - \lambda_2)(\lambda_1 - \lambda_2)(\lambda_3 - \lambda_2)} e^{-\lambda_2 t} \right. \\ \left. + \frac{(E_2 - \lambda_3)}{(k_a - \lambda_3)(\lambda_1 - \lambda_3)(\lambda_2 - \lambda_3)} e^{-\lambda_3 t} \right] \\ + D_3 \left[\frac{(E_1 - \lambda_1)(E_2 - \lambda_1)}{(\lambda_2 - \lambda_1)(\lambda_3 - \lambda_1)} e^{-\lambda_1 t} + \frac{(E_1 - \lambda_2)(E_2 - \lambda_2)}{(\lambda_1 - \lambda_2)(\lambda_3 - \lambda_2)} e^{-\lambda_2 t} \right. \\ \left. + \frac{(E_1 - \lambda_3)(E_2 - \lambda_3)}{(\lambda_1 - \lambda_3)(\lambda_2 - \lambda_3)} e^{-\lambda_3 t} \right] + \left[\frac{(J - D_1 k_{13} \lambda_1)}{(\lambda_1 - \lambda_2)(\lambda_1 - \lambda_3)} e^{-\lambda_1 t} \right. \\ \left. + \frac{(D_1 k_{13} \lambda_2 - J)}{(\lambda_1 - \lambda_2)(\lambda_2 - \lambda_3)} e^{-\lambda_2 t} + \frac{(D_1 k_{13} \lambda_3 - J)}{(\lambda_1 - \lambda_3)(\lambda_3 - \lambda_2)} e^{-\lambda_3 t} \right]$$

The results for the infusion type of compartment model are obtained from Fisher and Shafer (2007). For all other types of compartment models, the solutions are obtained from Abuhelwa, Foster, and Upton (2015).

CMPTMODEL Statement and Its Options

This section lists all required, conditionally required, and non-required options for the CMPTMODEL statement when you fit a one-, two-, or three-compartment model.

Table 1 One-Compartment Models

Model	Required Options	Conditionally Required Options	Non-required Options
One compartment with bolus dose	NCOMPS=1 ADMTYPE=IVB PARMTYPE=1	K10=	DOSE1= SCALE1=
One compartment with bolus dose	NCOMPS=1 ADMTYPE=IVB PARMTYPE=2	CL1=, VOL1=	DOSE1= SCALE1=
One compartment with infusion dose	NCOMPS=1 ADMTYPE=INF PARMTYPE=1	K10= RATE= or DURN=	DOSE1= SCALE1=
One compartment with infusion dose	NCOMPS=1 ADMTYPE=INF PARMTYPE=2	CL1=, VOL1= RATE= or DURN=	DOSE1= SCALE1=
One compartment with oral dose	NCOMPS=1 ADMTYPE=ORAL PARMTYPE=1	K10=, Ka=	DOSE1= SCALE1=
One compartment with oral dose	NCOMPS=1 ADMTYPE=ORAL PARMTYPE=2	CL1=, VOL1=, Ka=	DOSE1= SCALE1=

Table 2 Two-Compartment Models

Model	Required Options	Conditionally Required Options	Non-required Options
Two compartments with bolus dose	NCOMPS=2 ADMTYPE=IVB PARMTYPE=1	K10=, K12=, K21=	DOSE1=, DOSE2= SCALE1=, SCALE2= K20=, PCONC2=
Two compartments with bolus dose	NCOMPS=2 ADMTYPE=IVB PARMTYPE=2	CL1=, VOL1= CL2=, VOL2=	DOSE1=, DOSE2= SCALE1=, SCALE2= K20=, PCONC2=
Two compartments with infusion dose	NCOMPS=2 ADMTYPE=INF PARMTYPE=1	K10=, K12=, K21= RATE= or DURN=	DOSE1= SCALE1=
Two compartments with infusion dose	NCOMPS=2 ADMTYPE=INF PARMTYPE=2	CL1=, VOL1=, CL2=, VOL2= RATE= or DURN=	DOSE1= SCALE1=
Two compartments with oral dose	NCOMPS=2 ADMTYPE=ORAL PARMTYPE=1	K10=, K12=, K21= Ka=	DOSE1=, DOSE2= SCALE1=, SCALE2= K20=, PCONC2=
Two compartments with oral dose	NCOMPS=2 ADMTYPE=ORAL PARMTYPE=2	CL1=, VOL1=, CL2= VOL2=, Ka=	DOSE1=, DOSE2= SCALE1=, SCALE2= K20=, PCONC2=

Table 3 Three-Compartment Models

Model	Required Options	Conditionally Required Options	Non-required Options
Three compartments with bolus dose	NCOMPS=3 ADMTYPE=IVB PARMTYPE=1	K10=, K12=, K21= K13=, K31=	DOSE1=, DOSE2=, DOSE3= SCALE1=, SCALE2=, SCALE3= K20=, K30= PCONC2=, PCONC3=
Three compartments with bolus dose	NCOMPS=3 ADMTYPE=IVB PARMTYPE=2	CL1=, VOL1=, CL2= VOL2=, CL3=, VOL3=	DOSE1=, DOSE2=, DOSE3= SCALE1=, SCALE2=, SCALE3= K20=, K30= PCONC2=, PCONC3=
Three compartments with infusion dose	NCOMPS=3 ADMTYPE=INF PARMTYPE=1	K10=, K12=, K21= K13=, K31= RATE= or DURN=	DOSE1= SCALE1=
Three compartments with infusion dose	NCOMPS=3 ADMTYPE=INF PARMTYPE=2	CL1=, VOL1=, CL2= VOL2=, CL3=, VOL3= RATE= or DURN=	DOSE1= SCALE1=
Three compartments with oral dose	NCOMPS=3 ADMTYPE=ORAL PARMTYPE=1	K10=, K12=, K21= K13=, K31=, Ka=	DOSE1=, DOSE2=, DOSE3= SCALE1=, SCALE2=, SCALE3= K20=, K30= PCONC2=, PCONC3=
Three compartments with oral dose	NCOMPS=3 ADMTYPE=ORAL PARMTYPE=2	CL1=, VOL1=, CL2= VOL2=, CL3=, VOL3= Ka=	DOSE1=, DOSE2=, DOSE3= SCALE1=, SCALE2=, SCALE3= K20=, K30= PCONC2=, PCONC3=

ACKNOWLEDGMENTS

The authors are grateful to Yi Gong and Ed Huddleston of the Advanced Analytics Division at SAS Institute Inc. for their valuable assistance in the preparation of this manuscript.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author:

Raghavendra Rao Kurada
 SAS Institute Inc.
 SAS Campus Drive
 Cary, NC 27513
 919-531-2083
 Raghavendra.Kurada@sas.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.